Math 262a, Fall 1999, Glenn Tesler
Homework 9
12/10/99

Problem 2

Here's how I generated the sequence:

```
> rec := proc(n)
        option remember;
        (3*n^2+2*n-1)*rec(n-2)+(4*n-5)*rec(n-1);
  end:
  rec(0):=1: rec(1):=2:
> terms := ['rec(k)'$k=0..16];
```

terms := [1, 2, 21, 211, 3476, 69864, 1741060, 51222620, 1743410160, 67363596160, 2913873706640, 139508695484400, 7324686442350400, 418478888788169600, 25847105490242145600, 1716199939670189406400, 121907633727244649312000]

Normally you would find the start of the sequence from some empirical or brute force means.

```
>
> guessrec := proc(terms,I,J)
        local i,j,k,template,grec,eqs,sols,N,frec;

        k := nops(terms)-1;
        template :=
  sum(sum(a[i,j]*n^i*f(n+j),i=0..I),j=0..J);

        frec := unapply(terms[n+1],n); # make this a
  function
        grec := subs(f=frec,template);

        eqs := { seq(eval(subs(n=N,grec)),N=0..k-J) };
        sols := solve(eqs);
        if sols=NULL then RETURN(FAIL) fi;

        subs(sols,template) = 0;
  end:
> guessrec(terms,0,0);
```

$$0=0$$

```
> guessrec(terms,1,1);
```

$$0=0$$

```
> guessrec(terms,2,2);
```

$$-15\,a_{0,2}\,f(n)-14\,a_{0,2}\,n\,f(n)-3\,a_{0,2}\,n^2\,f(n)-3\,a_{0,2}\,f(n+1)-4\,a_{0,2}\,n\,f(n+1)+a_{0,2}\,f(n+2)=0$$

```
> simplify("/a[0,2]);
```

$$-15\,f(n)-14\,n\,f(n)-3\,n^2\,f(n)-3\,f(n+1)-4\,n\,f(n+1)+f(n+2)=0$$

```
> collect(",{f(n),f(n+1),f(n+2)});
```

$$f(n+2)+(-14\,n-3\,n^2-15)\,f(n)+(-4\,n-3)\,f(n+1)=0$$

Using too few terms gives spurious solutions:

```
> guessrec(terms[1..4],2,2);
```

$$(-2\,a_{0,1}-21\,a_{0,2})\,f(n)+a_{1,0}\,n\,f(n)+\left(-\frac{17}{2}\,a_{0,1}-\frac{169}{2}\,a_{0,2}-a_{1,0}-\frac{21}{2}\,a_{1,1}-\frac{21}{2}\,a_{2,1}-\frac{211}{2}\,a_{1,2}-\frac{211}{2}\,a_{2,2}\right)n^2\,f(n)+a_{0,1}\,f(n+1)+a_{1,1}\,n\,f(n+1)+a_{2,1}\,n^2\,f(n+1)+a_{0,2}\,f(n+2)+a_{1,2}\,n\,f(n+2)+a_{2,2}\,n^2\,f(n+2)=0$$

```
> guessrec(terms[1..9],2,2);
```

$$\left(-15\,a_{0,2}-\frac{4227789970168920}{1935065306831}\,a_{2,2}\right)f(n)+\left(-14\,a_{0,2}+\frac{5950106790518264}{1935065306831}\,a_{2,2}\right)n\,f(n)+\left(-3\,a_{0,2}+\frac{2479762766044862 8}{1935065306831}\,a_{2,2}\right)n^2\,f(n)+\left(-3\,a_{0,2}+\frac{2113894985084460}{1935065306831}\,a_{2,2}\right)f(n+1)+\left(-4\,a_{0,2}+\frac{488340789368264}{1935065306831}\,a_{2,2}\right)n\,f(n+1)+\frac{4197158147100063}{1935065306831}\,a_{2,2}\,n^2\,f(n+1)+a_{0,2}\,f(n+2)-\frac{930025877222492}{1935065306831}\,a_{2,2}\,n\,f(n+2)+a_{2,2}\,n^2\,f(n+2)=0$$

Using too high order/degree gives multiples of the desired recurrence with lots of a[i,j]'s still free:

```
> bigrec := guessrec(terms,3,3);
```

$$bigrec := (-15\, a_{0,2} - 105\, a_{0,3})\, f(n) + (160\, a_{1,3} + 6\, a_{0,2} + 142\, a_{0,3} + 5\, a_{1,1})\, n\, f(n) + \left(\frac{47}{3} a_{0,2} + 203\, a_{0,3} + \frac{448}{3} a_{1,3} + \frac{14}{3} a_{1,1}\right) n^2\, f(n) + (32\, a_{1,3} + 4\, a_{0,2} + 48\, a_{0,3} + a_{1,1})\, n^3\, f(n) + (-3\, a_{0,2} - 53\, a_{0,3})\, f(n+1) + a_{1,1}\, n\, f(n+1) + \left(\frac{68}{3} a_{1,3} + \frac{16}{3} a_{0,2} + 61\, a_{0,3} + \frac{4}{3} a_{1,1}\right) n^2\, f(n+1) - 3\, a_{1,3}\, n^3\, f(n+1) + a_{0,2}\, f(n+2) + \left(-\frac{53}{3} a_{1,3} - \frac{4}{3} a_{0,2} - 20\, a_{0,3} - \frac{1}{3} a_{1,1}\right) n\, f(n+2) - 4\, a_{1,3}\, n^2\, f(n+2) + a_{0,3}\, f(n+3) + a_{1,3}\, n\, f(n+3) = 0$$

The distinction between the last two things is that in bigrec, the number of equations k-J+1 greatly exceeds the number of unknowns, (I+1)(J+1), yet is still solvable; but with too few terms, the number of equations is smaller than the number of unknowns, and of course there's a solution, but it's meaningless due to insufficient data.

In bigrec, the coefficient of each a[i,j] is a recurrence:

```
> as := indets(bigrec,indexed);
```

$$as := \{a_{0,2}, a_{0,3}, a_{1,3}, a_{1,1}\}$$

```
> map(collect,[coeffs(expand(lhs(bigrec)),as)],{f(n),f(n+1
  ),f(n+2),f(n+3)},factor);
```

$$\left[\left(1 - \frac{4}{3} n\right) f(n+2) + \frac{1}{3} (3\, n + 5)\, (n+3)\, (4\, n - 3)\, f(n) + \frac{1}{3} (4\, n - 3)\, (4\, n + 3)\, f(n+1),\ -20\, n\, f(n+2) + (3\, n + 5)\, (16\, n - 7)\, (n+3)\, f(n) + (-53 + 61\, n^2)\, f(n+1) + f(n+3),\ -\frac{1}{3} n\, (12\, n + 53)\, f(n+2) + \frac{32}{3} n\, (n+3)\, (3\, n + 5)\, f(n) - \frac{1}{3} n^2\, (-68 + 9\, n)\, f(n+1) + n\, f(n+3),\ -\frac{1}{3} n\, f(n+2) + \frac{1}{3} n\, (n+3)\, (3\, n + 5)\, f(n) + \frac{1}{3} n\, (4\, n + 3)\, f(n+1)\right]$$

Express them in operator notation and take their GCD to get the optimal recurrence.

```
>
```

```
>
```